
PWL_Noise_OTA
**
*          _____________________
*         |  VCC      |        _|_
*         |           |       /VCC\
*          ->       <-        \___/
*        XQP1`|___|'QP2        _|_
*          _ '| | |`_          ///
*         |   : |____|VTA2                     _____
*         |  VC2     |                        |     |-> V_AVE
*         |         <-                   OUT->|M_RMS|
*         |_______|'XQP3                      |_____|-> V_RMS
*         |VTA1   |`_
*         |       :  | OUT
*         |      VC3 |_____/\  /\  /\_        :     :     :
*         |          |       \/  \/   |      _|_   _|_   _|_
*         |          |         R1    _|_    /VC1\ /VC2\ /VC3\
*        _|          |_              ///    \___/ \___/ \___/
*    __|'XQN1      QN2 `|__                   |     |     |
*  _|_ |`->   VE     <-'| _|_                _|_   _|_   _|_
*  /// :  |__________|    ///                ///   ///   ///
*     VC1   |                          Noise Control voltages
*           |      ____________
*           |     | VB  |     _|_                    _____
*           |_    |    _|    / _ \                 _|_   _|_
*         XQN3`|__|__|'QN4   \/ \/IB              /VEE\  ///
*           <-'|     |`->    /\_/\                \___/
*           |           |    \___/                 _|_
*          _|_         _|_    _|_                  \ /
*          \ /         \ /    \ /                   V VEE
*           V  VEE      V      V
*

.OPTIONS  GMIN=1e-12   METHOD=trap   srcsteps = 100  gminsteps = 1

.OPTIONS  ITL1=400     ITL6=100
*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VCC    0             DC     2
VEE       0      VEE           DC     2
IB1       VE     0             DC     1u
XQN1      VTA1   0      VE     VC1    XQN1X
QN2       OUT    0      VE     NPN1   1.00 

XQP1      VTA1   VTA2   VCC    VC2    XQP1X
QP2       VTA2   VTA2   VCC    PNP1   1
XQP3      OUT    VTA1   VTA2   VC3    XQP1X
R1        OUT    0                    1
VC1       VC1    0             DC     1
VC2       VC2    0             DC     0
VC3       VC3    0             DC     0
C1        VTA1   0             10pf

Vstart    VS     0             Pulse  (0 1 1m 1n 1n 10s 20s)
B2        OUT2   0             V = v(OUT)*v(VS)+1p
XM_RMS    OUT2   V_AVE  V_RMS  M_RMS
.tran    5u                 50m        0          5u         UIC

**#0====All_Matched_Transistors_In_An_OTA_Add_Noise===

.control

**#1=====First_Observe_Noise_IN_XQN1_Alone============
run
plot      out2 v_ave v_rms    ylimit -.5n .5n       title    QN1

**#2=====Next_Observe_Noise_IN_XQP1_Alone=============
alter     vc1            dc = 0
alter     vc2            dc = 1
run
plot      out2 v_ave v_rms    ylimit -.5n .5n       title    QP1

**#2=====Last_Observe_Noise_IN_XQP3_Alone=============
alter     vc2            dc = 0
alter     vc3            dc = 1
run
plot      out2 v_ave v_rms    ylimit -.5n .5n       title    QP3
.endc

*====== .models Using Near Perfect Transistors ============
.model     NPN1   NPN( BF = 1MEG VAF =  1MEG )
.model    PNP1   PNP( BF = 1MEG VAF =  1MEG )

*====== .SUBCKT  XQN1X Models Audio Noise to NPN current ============
*                  ___
*                 |(C)|
*                 |___|    OUT_1KOHM    
*                   |        ^
*  ____    ___ B1  _|       /_\         
*  |(B)|__/BN \__|'          |          Inoise_collector = sqrt(2*q*Idc*BW)
*  |___|  \___/  |`-> E1    _|__  PWL   Vnoise_base_rb@0 = Inoise*re
*           :      _|_     /_   \                        = Inoise*(Kt/q)/Idc
*          _:_    /VM \   // \   \                      
*         |NGN|   \___/   \   \_//
*         |___|     |      \____/  .include Noise_audio1K.txt
*                  _|_       |       
*   NoiseGainNode |(E)|     _|_
*   1  = Normal   |___|     ///
*   0  = Off
*   10 = 10X gain



.SUBCKT  XQN1X  C   B   E    NCN

.include                               Noise_audio1K.txt
QN1           C     B1     E1    NPN1   1
VM            E1    E      DC    0

BN            B     B1     V =   v(NCN)*V(audio1K)*(13u/(i(VM)+5n))^.5
.ENDS    XQN1X

*====== .SUBCKT  XQP1X Models Audio Noise to PNP current ============
*                  ___
*                 |(C)|
*                 |___|    OUT_1KOHM    
*                   |        ^
*  ____    ___ B1  _|       /_\         
*  |(B)|__/BN \__|'          |          Inoise_collector = sqrt(2*q*Idc*BW)
*  |___|  \___/  |`<- E1    _|__  PWL   Vnoise_base_rb@0 = Inoise*re
*           :      _|_     /_   \                        = Inoise*(Kt/q)/Idc
*          _:_    /VM \   // \   \                      
*         |NGN|   \___/   \   \_//
*         |___|     |      \____/  .include Noise_audio1K.txt
*                  _|_       |       
*   NoiseGainNode |(E)|     _|_
*   1  = Normal   |___|     ///
*   0  = Off
*   10 = 10X gain

.SUBCKT  XQP1X  C   B   E    NCN

.include                               Noise_audio1K2.txt
QP1           C     B1     E1    PNP1   1
VM            E1    E      DC    0

BN            B     B1     V =   v(NCN)*V(audio1K2)*(13u/(i(VM)+5n))^.5
.ENDS    XQP1X

* ====== .SUBCKT M_RMS Sanity Check RMS Meter =======
*              ____
*             |    |-> V_AVE
*       V_IN->| RMS|            A Sanity RMS Meter
*             |____|-> V_RMS
*
*
*          Vsum    Rave         Vsum2   Rave2       V_RMS    Rrms
*          __/\  /\  /\_ V_AVE  __/\  /\  /\_ Vave2 __/\  /\  /\_
*        _|_   \/  \/   |     _|_   \/  \/   |    _|_   \/  \/   |
* V_IN->/   \          _|_   /   \          _|_  /   \          _|_
*       \___/Bsum      ___   \___/Bsum2     ___  \___/Brms      ///
*         |             |      |             |     |
*        _|_      Cave _|_    _|_     Cave2 _|_   _|_
*        ///           ///    ///           ///   ///
*

.SUBCKT M_RMS  V_IN  V_AVE V_RMS
Bsum       Vsum      0         V = v(V_IN)
Rave         Vsum      V_AVE  1
Cave         V_AVE  0         10m   
Bsum2     Vsum2     0         V = (v(V_IN) -v(V_AVE))^2
Rave2        Vsum2     Vave2     1
Cave2        Vave2     0         7m
Brms       V_RMS  0         V = v(Vave2)^0.5
Rrms         V_RMS  V0        1k
.ENDS    M_RMS

.end

=====================END_OF_SPICE============================

**#1=====First_Observe_Noise_IN_XQN1_Alone============
run
plot      out2 v_ave v_rms    ylimit -.5n .5n       title    QN1

**#2=====Next_Observe_Noise_IN_XQP1_Alone=============



alter     vc1            dc = 0
alter     vc2            dc = 1
run
plot      out2 v_ave v_rms    ylimit -.5n .5n       title    QP1

**#3=====Last_Observe_Noise_IN_XQP3_Alone=============
alter     vc2            dc = 0
alter     vc3            dc = 1
run
plot      out2 v_ave v_rms    ylimit -.5n .5n       title    QP3

Only the matched pairs add noise
In the LM3080 architecture, 
eight transistors all add equal shot noise.
         ^
        /_\    LM3080
       __|_______________________   ___
      |           |  |           | |   | Output
       ->       <-    ->       <-  |___|
      PNP\|___|/PNP  PNP\|___|/PNP   |
        _'| | |`_      _'| | |`_     |
       |    |____|    |    |____|    |
       |         |    |         |    |
       |        <-    |        <-    |
       |______|/PNP   |______|/PNP   |
Control|      |`_     |      |`______|
Current|         |    |              |
   ^   |         |____\ ________     |
  /_\  |_______       /         |    |
  _|_          |      |         |    |
 / _ \         |      |         |    |
 \/ \/   _    _|      |__    _  |    |
 /\_/\  | |_|'  NPN NPN  `|_| | |    |
 \___/  |_| |`->       <-'| |_| |    |
   |    -IN    |_______|    +IN |    |
   |             |              |    |
  _|_           _|              |   _|
 |   |________|'  NPN      _____|_|'  NPN
 |___| |      |`->         |      |`->
 IABI  |     ____|         |     ____|
       |_   |   _|         |_   |   _|
    NPN  `|_|_|'        NPN  `|_|_|'  NPN
       <-'|   |`->         <-'|   |`->
       |_________|         |_________|
          _|_                  _|_
          ///                  ///
          Gnd                  Gnd




